



**Western Region Technical Attachment
No. 94-13
April 5, 1994**

UNRAVELING THE PCGRIDS MACRO MYSTERY

Keith Meier - WRH SSD - Salt Lake City, UT

Introduction

The use of gridded data and PCGRIDS has grown rapidly within the Western Region, as well as nationally. The sophistication of PCGRIDS users ranges all the way from the novice, who understands the basic command structure and is able to display fields of interest using the menu system (CMD.EXE), to the expert, who is able to create more complicated macros that incorporate some of the latest science. Many users fall somewhere in between these two extremes.

The intent of this Technical Attachment is to provide background knowledge on macro "logic" and macro creation, which will assist in continuing development of PCGRIDS expertise. Macro basics will be discussed first. This will provide the foundation necessary for the second section which discusses the generation of more complicated PCGRIDS macros. This will be accomplished through the use of both simple and complicated examples.

Macro Basics

PCGRIDS macros provide a mechanism for generating fields used frequently without retyping long command strings. Once commands have been correctly edited into a macro file (*.cmd), macros may be initiated by a few keystrokes from the menu program. This provides some stability in viewing fields day after day in relation to contour intervals, color schemes, and the combinations of fields. In general, macros replicate those commands that are issued interactively on the command line. Various commands are especially suited for use within PCGRIDS macros, such as LOOP, STON, STOF, ENDL, TXT#, and #SEC. The LOOP command alerts PCGRIDS that the subsequent lines are to be executed sequentially without user intervention, until an ENDL command is encountered. For example,

```
LOOP
STOF
HIGHT CI60 CLR3/WVRT 500 DASH C2-5 F00
TXT2 500 mb Heights (dam) & Absolute Vorticity at 00 hr
STON
ENDL
```

will initially deactivate the listing of the maximum, minimum, mean, and standard deviation values of each plotted field (STOF). 500 mb heights (in color 3) and absolute vorticity (dashed) will be plotted for the 00 hr forecast using 60 m and $2 \times 10^{-5} \text{ s}^{-1}$ contour intervals,

respectively. A text line will then be plotted on the second line from the top of the screen stating, "500 mb Heights (dam) & Absolute Vorticity at 00 hr", after which the statistics are activated (STON). When PCGRIDS encounters the ENDL command, it will cease to execute commands within a macro until a carriage return is initiated by the user.

Time pauses may be included in macros to simulate a looping feature, such as 3SEC, which would pause 3 seconds before executing the next command string in the macro. For example,

```
LOOP
STOF
HGHT CI60 CLR3 500 F00
5SEC
WVRT C2-5 DASH/
TXT2 500 mb Heights (dam) & Absolute Vorticity at 00 hr
STON
ENDL
```

accomplishes the same as the first example, but will pause 5 seconds between finishing the plot of 500 mb heights and beginning the plot of 500 mb absolute vorticity.

NOTE: A macro can not be executed from within another macro! For example, using the time-section labeling macro, TLBL.CMD, within a macro which creates a time-section of relative humidity will fail. Thus,

```
LOOP
RELH CI10
TLBL.
ENDL
```

will never work. Instead, the commands contained within TLBL.CMD should be incorporated on the lines following RELH CI10. This would result in a time-section of relative humidity with an overlay of the time section labels. Also, potential temperature surfaces cannot be created within a macro, using the MTHT command.

The information above will assist many people in writing basic macros. More complicated meteorological parameters such as frontogenesis, deformation, and Q-vectors require more detailed background information, provided below.

Creating Complicated Macros

With meteorological research demonstrating the applicability of a growing number of computed quantities, the functionality of gridded data is increased by the ability to manipulate the data into meaningful displays of these "new" (or previously unused) meteorological fields and parameters. For example, Q-vectors have been used in the research community for a number of years, but the inability to display forecast Q-vector fields operationally has limited their use. With PCGRIDS, it is possible to string together the appropriate PCGRIDS commands to generate forecast Q-vector fields. Of course, there are many other fields and parameters that may now be calculated, viewed, and used in real-time

in the forecast process.

The logic necessary to create PCGRIDS macros, which compute these derived fields, is not necessarily intuitive. As most PCGRIDS users are aware, PCGRIDS processes command lines from right to left. Thus, the four-letter commands the program encounter first (from the right) are executed first. The command line, "SSUM DSDX THTA DSDY THTA," initially computes DSDY THTA, then DSDX THTA, and finally adds the two together. To accomplish this, PCGRIDS places processed fields on what can be thought of as "storage stacks". For example, DSDY THTA is calculated and placed on the first position of the storage stack. DSDX THTA is then calculated and placed on the first position of the storage stack, moving the field DSDY THTA to the second position of the storage stack. Now, when PCGRIDS encounters the SSUM command, the program realizes that it needs two fields to complete the sum operation. Thus, it searches the last two storage stacks for the fields to add together. When it locates and adds them, the plotted field appears on the screen. Understanding this logic is crucial to creating more complicated macros that require multiple command lines to complete.

One of the more complicated calculations, which demonstrates a variety of macro creation concepts, is Pettersen's two-dimensional frontogenesis equation, as defined below.

$$-\frac{1}{|\nabla T|} \left(\frac{\delta\theta}{\delta x} \frac{\delta\theta}{\delta x} \frac{\delta u}{\delta x} + \frac{\delta\theta}{\delta x} \frac{\delta\theta}{\delta y} \frac{\delta v}{\delta x} + \frac{\delta\theta}{\delta x} \frac{\delta\theta}{\delta y} \frac{\delta u}{\delta y} + \frac{\delta\theta}{\delta y} \frac{\delta\theta}{\delta y} \frac{\delta v}{\delta y} \right)$$

Term		A	B	C	D
Line	7	3	4	5	6

The following macro commands will generate this calculation. A detailed, line-by-line discussion of the macro will follow to illustrate the "logic" of more complicated PCGRIDS macros.

- 1) LOOP
- 2) STOF
- 3) CONT SMLT DSDX THTA SMLT DSDX THTA DSDX XCMP WIND F00
- 4) CONT SSUM SMLT DSDX THTA SMLT DSDY THTA DSDX YCMP WIND KEPS
- 5) CONT SSUM SMLT DSDX THTA SMLT DSDY THTA DSDY XCMP WIND KEPS
- 6) CONT SSUM SMLT DSDY THTA SMLT DSDY THTA DSDY XCMP WIND KEPS
- 7) SMLC -10. SMLT INVS MGRD TEMP KEPS DNEG C1-9
- 8) ENDL

Line 1 simply tells PCGRIDS to execute the macro commands one after the other until an ENDL is encountered (as discussed previously). Line 2 prevents the statistics from being plotted to the screen. Sometimes plotting these statistics assists in determining contour intervals or the appropriate magnitude of the quantity.

Lines 3 through 7 are directly involved in the computation of the equation above. Keep in mind the principles of the "storage stacks" discussed earlier. Reading right to left (as PCGRIDS does), line 3: sets PCGRIDS to the 00 hr forecast; calculates DSDX XCMP WIND (or DSDX of the u-component of the observed wind); calculates DSDX THTA and multiplies

(SMLT) it by the field previously computed (DSDX XCMP WIND); calculates DSDX THTA and multiplies it by the field previously computed (SMLT DSDX THTA DSDX XCMP WIND). The CONT command alerts PCGRIDS that the command string continues onto another line and the previously computed field (SMLT DSDX THTA SMLT DSDX THTA DSDX XCMP WIND) is placed on the storage stack for later use. This is term A in the equation.

Line 4 uses the command KEPS to alert PCGRIDS that this command string will be retrieving a scalar field from the storage stack (if a vector was to be retrieved, KEPV would access the two scalar fields on the storage stack that are the components of the vector field). Line 4 continues by calculating DSDX YCMP WIND, multiplying it by DSDY THTA, multiplying this result by DSDX THTA (resulting in term B of the equation), and finally adding SMLT DSDX THTA SMLT DSDY THTA DSDX YCMP WIND to the quantity calculated in line 3, and places it on the storage stack for use in the next line (as indicated by the CONT command).

Line 5 calculates term C of the equation and adds it to the result of previously added terms A and B (residing on the storage stack). Line 6 likewise calculates term D of the equation and adds it to the previous sum of terms A, B, and C, which were residing on the storage stack. Now, the sum of terms A, B, C, and D are placed on the storage stack for use in the next command line (due to the presence of the CONT command).

Line 7 begins by setting the contour interval to $1 \times 10^{-9} \text{ K s}^{-1} \text{ m}^{-2}$, sets negative-valued contours to dash, and activates the scalar field on the storage stack. Line 7 continues by calculating the magnitude of the temperature gradient, taking the inverse of that field, multiplying it by the scalar field on the storage stack (the sum of terms A, B, C, and D), multiplying this quantity (essentially the result of the equation) by -10. This is necessary since the result of the frontogenesis equation is on the order of 10^{-10} , and PCGRIDS does not allow contour intervals to be set to magnitude. Thus, multiplying by 10 artificially scales the frontogenesis by an order of magnitude [$10 * 10^{-10} = 10^{-9}$]. Now, since the CONT does not begin the command line, PCGRIDS will plot the results of the last calculation onto the screen, which is Pettersen's two-dimensional frontogenesis equation.

Many other macros can be modeled after the above example, by successively combining calculated fields. Although this may seem like a cumbersome way to calculate complicated fields, creating macros of this sort become easier with practice. It is always helpful to scrutinize the resulting plot to be sure you have calculated the field you intended. By experimentation and idea sharing, the "logic" of PCGRIDS command macros will become more apparent.